



## **Progetto di Ricerca Operativa**

*Ottimizzazione dei Costi per i Servizi di Trasporto Ordinario  
e di Emergenza*

Soldà Matteo  
Matricola 1226319

A.A. 2022–2023

# Contents

<b>1</b>	<b>Introduzione</b>	<b>3</b>
1.1	Abstract . . . . .	3
1.2	Problema Generale . . . . .	3
<b>2</b>	<b>Modello</b>	<b>4</b>
2.1	Insiemi . . . . .	4
2.2	Parametri . . . . .	4
2.3	Variabili . . . . .	4
2.4	Funzione Obiettivo . . . . .	4
<b>3</b>	<b>Codice AMPL</b>	<b>6</b>
3.1	File <i>.mod</i> . . . . .	6
<b>4</b>	<b>Scenari</b>	<b>8</b>
4.1	Primo Scenario . . . . .	8
4.1.1	Descrizione . . . . .	8
4.1.2	Dati . . . . .	8
4.1.3	Output . . . . .	8
4.2	Secondo Scenario . . . . .	9
4.2.1	Descrizione . . . . .	9
4.2.2	Output . . . . .	9
4.3	Terzo Scenario . . . . .	9
4.3.1	Descrizione . . . . .	9
4.3.2	Output . . . . .	9
<b>5</b>	<b>Extra</b>	<b>10</b>
5.1	Output su File . . . . .	10
5.2	Stampa Personalizzata per Problema Non Risolvibile . . . . .	10

# 1 Introduzione

## 1.1 Abstract

La Centrale Operativa del 118 della città di Padova necessita di un gran numero di ambulanze per svolgere i trasporti sia ordinari che di emergenza nell'intera provincia, ma con le ambulanze interne e quelle in possesso degli ospedali periferici si riesce a coprire solo in minima parte il fabbisogno: si rende quindi necessario rivolgersi alle Organizzazioni di Volontariato che, in cambio di un rimborso spese, prestano ambulanze e soccorritori per lo svolgimento degli interventi. L'obiettivo del progetto è quello di minimizzare il costo per il mantenimento del servizio, garantendo comunque un pronto intervento in caso di necessità.

## 1.2 Problema Generale

Ogni giorno la Centrale Operativa del 118 di Padova e l'Azienda Ospedaliera ricevono una moltitudine di chiamate, sia per richieste di supporto immediato (che vengono gestite dal SUEM), sia per trasporti programmati (che passano per le linee dell'Azienda Ospedaliera e che non rappresentano casi di emergenza, ma dedicati al trasporto di persone invalide presso l'ospedale per le visite programmate oppure per le dimissioni dal Pronto Soccorso).

L'Azienda Ospedaliera, al suo interno, dispone di un determinato numero di ambulanze di tipo B (dedicate ai trasporti), mentre quelle di tipo A (dedicate al SUEM) sono fornite dagli ospedali periferici. L'insieme di questi mezzi non è però sufficiente a coprire le richieste pervenute dall'intera provincia, ma ne ricopre una minima parte. Per questo motivo, la Centrale Operativa si rivolge a tre Organizzazioni di Volontariato (OdV) che offrono il loro supporto per coprire i viaggi scoperti:

- La Croce Rossa può fornire solo ambulanze di tipo A
- La Croce Bianca può fornire solo ambulanze di tipo B
- La Croce Verde può fornire sia ambulanze di tipo A che di tipo B

Il numero di ambulanze complessive in un determinato giorno viene stimato rispetto alla media di interventi e trasporti effettuati negli stessi giorni delle settimane precedenti, mantenendo comunque un numero arbitrario di ambulanze sia di tipo A che di tipo B in più.

Si vuole quindi minimizzare il costo complessivo settimanale per il servizio, sapendo che:

- Ogni giorno la Centrale Operativa ha necessità di un quantitativo diverso di ambulanze
- Qualora si attivasse una OdV per l'assistenza settimanale, è prevista una singola quota fissa di attivazione, indipendentemente dal tipo di ambulanza attivata
- Ogni ambulanza richiamata dalle OdV prevede un rimborso spese giornaliero che varia in base all'organizzazione e al tipo di ambulanza attivata
- Le ambulanze dell'Azienda Ospedaliera e degli ospedali periferici, se attivate in un determinato giorno, prevedono dei piccoli costi di manutenzione
- Ogni OdV e Ospedale dispone di un numero limitato di ambulanze
- Ogni giorno bisogna attivare un numero arbitrario di ambulanze in più rispetto al fabbisogno
- L'Ospedale deve attivare un minimo di fornitori ogni giorno
- Le OdV possono intervenire per un numero limitato di giorni

**Disclaimer:** questo progetto riguarda un tema reale con dati fittizi. Le Organizzazioni di Volontariato citate nell'introduzione e l'Ospedale di Padova sono realmente esistenti, ma i dati riguardanti disponibilità di mezzi, stima della necessità, metodologie di intervento e costi sostenuti sono casuali e adattati per rendere consistente lo studio in questione. Ogni riferimento è quindi da ritenersi puramente casuale in quanto frutto dell'immaginazione del redattore.

## 2 Modello

### 2.1 Insiemi

- *Giorni*: giorni della settimana in cui è attivo il servizio
- *Fornitori*: fornitori che possono fornire ambulanze
- *Tipo*: tipi di ambulanze disponibili (indica la destinazione d'uso del mezzo nel contesto)

### 2.2 Parametri

- $bisogno_{g,t}$ : numero di ambulanze di tipo  $t \in Tipo$  necessarie per il giorno  $g \in Giorni$
- $surplus_{g,t}$ : numero di ambulanze di scorta di tipo  $t \in Tipo$  necessarie per il giorno  $g \in Giorni$
- $maxAmbulanze_{t,f}$ : numero massimo di ambulanze di tipo  $t \in Tipo$  che il fornitore  $f \in Fornitori$  può fornire al giorno
- $maxGiorni_{t,f}$ : numero massimo di giorni in cui un fornitore  $f \in Fornitori$  può fornire ambulanze di tipo  $t \in Tipo$
- $costoGiornaliero_{t,f}$ : costo giornaliero per l'utilizzo di un'ambulanza di tipo  $t \in Tipo$  del fornitore  $f \in Fornitori$
- $addSurplus_f$ : aumento percentuale del costo per l'utilizzo di un'ambulanza del fornitore  $f \in Fornitori$  come surplus
- $costoAttivazioneSettimanale_f$ : costo di attivazione del fornitore  $f \in Fornitori$  per la settimana
- $BigM$ : costante sufficientemente grande utilizzata per i vincoli logici
- $MIN\_FORNITORI$ : costante che indica il numero minimo di fornitori da attivare al giorno

### 2.3 Variabili

- $ambulanze_{t,f,g}$ : numero di ambulanze di tipo  $t \in Tipo$  del fornitore  $f \in Fornitori$  attivate il giorno  $g \in Giorni$
- $ambulanzeSurplus_{t,f,g}$ : numero di ambulanze di tipo  $t \in Tipo$  del fornitore  $f \in Fornitori$  attivate il giorno  $g \in Giorni$  come surplus
- $attivazioneSettimanale_f = \begin{cases} 1 & \text{se viene attivato il fornitore } f \text{ per la settimana} \\ 0 & \text{altrimenti} \end{cases}$
- $attivazioneGiornaliera_{t,f,g} = \begin{cases} 1 & \text{se viene attivata l'ambulanza di tipo } t \text{ del fornitore } f \text{ il giorno } g \\ 0 & \text{altrimenti} \end{cases}$

### 2.4 Funzione Obiettivo

La funzione obiettivo del problema può essere espressa in forma generale come segue:

$$\min \quad \text{Costo Giornaliero Ambulanze Standard} + \text{Costo Giornaliero Ambulanze Surplus} \\ + \text{Costo Attivazione Settimanale} + \text{Costo Attivazione Giornaliero}$$

Dove i termini sono così definiti:

$$\text{Costo Giornaliero Ambulanze Standard} = \sum_{t \in Tipo, f \in Fornitori, g \in Giorni} ambulanze[t, f, g] \cdot costoGiornaliero[t, f]$$

$$\text{Costo Giornaliero Ambulanze Surplus} = \text{Costo Giornaliero Ambulanze Standard} + \text{CostoGiornalieroAmbulanzeStandard} \cdot addSurplus[t, f]$$

$$\text{Costo Attivazione Settimanale} = \sum_{f \in Fornitori} attivazioneSettimanale[f] \cdot costoAttivazioneSettimanale[f]$$

$$\text{Costo Attivazione Giornaliera} = \sum_{t \in Tipo, f \in Fornitori, g \in Giorni} attivazioneGiornaliera[t, f, g] \cdot costoGiornaliero[t, f]$$

**subject to**

- Ogni giorno la Centrale Operativa ha necessità di un numero diverso di ambulanze:

$$\sum_{t \in Tipo, f \in Fornitori} ambulanze[t, f, g] \geq bisogno[t, g] \quad \forall g \in Giorni$$

$$\sum_{t \in Tipo, f \in Fornitori} ambulanzeSurplus[t, f, g] \geq surplus[t, g] \quad \forall g \in Giorni$$

- I fornitori dispongono di un numero massimo di ambulanze fornibili in un giorno:

$$ambulanze[t, f, g] + ambulanzeSurplus[t, f, g] \leq maxAmbulanze[t, f] \quad \forall t \in Tipo, f \in Fornitori, g \in Giorni$$

- Le ambulanze di un determinato fornitore si possono attivare se e solo se si è pagata la quota di attivazione settimanale:

$$ambulanze[t, f, g] + ambulanzeSurplus[t, f, g] \leq BigM \cdot attivazioneSettimanale[f]$$

- Ogni giorno bisogna attivare un numero minimo di fornitori:

$$\sum_{f \in Fornitori} attivazioneGiornaliera[t, f, g] \geq MIN\_FORNITORI \quad \forall t \in Tipo, g \in Giorni$$

$$attivazioneGiornaliera[t, f, g] \leq attivazioneSettimanale[f] \quad \forall t \in Tipo, f \in Fornitori, g \in Giorni$$

$$ambulanze[t, f, g] + ambulanzeSurplus[t, f, g] \leq BigM \cdot attivazioneGiornaliera[t, f, g]$$

$$\forall t \in Tipo, f \in Fornitori, g \in Giorni$$

- Ogni fornitore può essere attivato un numero massimo di volte durante la settimana

$$\sum_{g \in Giorni} attivazioneGiornaliera[t, f, g] \leq maxGiorni[t, f] \quad \forall t \in Tipo, f \in Fornitori$$

**Domini:**

- $ambulanze_{t,f,g} \geq 0$
- $ambulanzeSurplus_{t,f,g} \geq 0$
- $attivazioneSettimanale_f \in [0, 1]$
- $attivazioneGiornaliera_{t,f,g} \in [0, 1]$

## 3 Codice AMPL

### 3.1 File *.mod*

**N.B.:** i commenti contenenti *Vincolo  $x.y$* , indicano quale vincolo richiesto via mail (e visibile tramite la scheda *issues* della repository GitHub del progetto) venga soddisfatto.

---

```
# =====
#      Modello AMPL
#      Ricerca operativa
#      Solda' Matteo
#      (2022)
# =====

### INSIEMI ###
set Giorni ordered;
set Fornitori ordered;
set Tipo; # Vincolo 3

### PARAMETRI ###
# Numero di Ambulanze Necessarie
param bisogno{ Giorni, Tipo } >= 0 integer default 7;

# Numero di Ambulanze di Scorta
param surplus{ Giorni, Tipo } >= 0 integer default 2;

# Numero Massimo di Ambulanze di un Determinato tipo Fornibili da un
  Determinato Fornitore
param maxAmbulanze{ Tipo, Fornitori } >= 0 integer default 5;

# Numero Massimo di Giorni in cui un Determinato Fornitore puo' Essere
  Attivato (Separatamente per Tipo di Ambulanza) (Vincolo 1)
param maxGiorni{ Tipo, Fornitori } >= 0 <= 7 integer default 7;

# Costo Giornaliero per una Ambulanza di un Determinato Tipo di un
  Determinato Fornitore
param costoGiornaliero{ Tipo, Fornitori } >= 0 default 15;

# Aumento Percentuale per Servizio in Surplus (Vincolo 2.2)
param addSurplus{ Fornitori } >= 0 default 0.50;

# Costo di Attivazione Settimanale per l'Attivazione di un Fornitore
  Indipendentemente dal Tipo di Ambulanza
param costoAttivazioneSettimanale{ Fornitori } >= 0 default 100;

# BigM per Vincoli di Tipo Logico (COSTANTE)
param BigM > 0 integer default 500;

# Numero Minimo di Fornitori da Attivare al Giorno (COSTANTE) (Vincolo 2.3)
param MIN_FORNITORI > 0 <= card(Fornitori) default (card(Fornitori)/2);

### VARIABILI ###
# Ambulanze di Tipo t del Fornitore f Attivate il Giorno g (+ Vincolo 2.2)
var ambulanze{ t in Tipo, f in Fornitori, g in Giorni } >= 0 integer;
var ambulanzeSurplus{ t in Tipo, f in Fornitori, g in Giorni } >= 0
  integer;

# Variabile Logica per l'Attivazione Settimanale del Fornitore f
  Indipendentemente dal Tipo di Ambulanza (Vincolo 2.1)
```

```

var attivazioneSettimanale{ f in Fornitori } binary;

# Variabile Logica per l'Attivazione Giornaliera di un Determinato
  Fornitore in un Determinato Giorno (Indistintamente dal Tipo di
  Ambulanza) (Vincolo 2.3)
var attivazioneGiornaliera{ t in Tipo, f in Fornitori, g in Giorni } binary;

### FUNZIONE OBIETTIVO ###
minimize costo:
  (sum{ t in Tipo, f in Fornitori, g in Giorni } ambulanze[t, f, g] *
    costoGiornaliero[t, f]) +
  (sum{ t in Tipo, f in Fornitori, g in Giorni } (ambulanzeSurplus[t, f,
    g] * costoGiornaliero[t, f] +
    (ambulanzeSurplus[t, f, g] * costoGiornaliero[t, f] * addSurplus[f]))) +
  (sum{ f in Fornitori } attivazioneSettimanale[f] *
    costoAttivazioneSettimanale[f]) +
  (sum{ t in Tipo, f in Fornitori, g in Giorni } attivazioneGiornaliera[t,
    f, g] * costoGiornaliero[t, f])
;

### VINCOLI ###
# Vincolo per la Necessita' Giornaliera (+ Vincolo 2.1)
subject to necessitaGiornaliera { t in Tipo, g in Giorni } : sum{ f in
  Fornitori } ambulanze[t, f, g] >= bisogno[g, t];
subject to necessitaGiornalieraSurplus { t in Tipo, g in Giorni } : sum{ f
  in Fornitori } ambulanzeSurplus[t, f, g] >= surplus[g, t];
subject to maxDisponibilita { t in Tipo, f in Fornitori, g in Giorni } :
  ambulanze[t, f, g] + ambulanzeSurplus[t, f, g] <= maxAmbulanze[t, f];

# Vincolo Logico per l'Attivazione Settimanale di un Fornitore
  Indipendentemente dal Tipo di Ambulanza Attivata (Vincolo 2.1)
subject to attivazioneSettimanaleFornitore { t in Tipo, f in Fornitori, g
  in Giorni } : ambulanze[t, f, g] <= BigM * attivazioneSettimanale[f];

# Vincolo Logico per l'Attivazione di Almeno MIN_FORNITORI Fornitori in un
  Giorno (Considerando il Tipo di Ambulanza) (Vincolo 2.3) + Vincoli
  Derivanti
subject to attivazioneMinima { t in Tipo, g in Giorni } : sum{ f in
  Fornitori } attivazioneGiornaliera[t, f, g] >= MIN_FORNITORI;
subject to collegamento2 { t in Tipo, f in Fornitori, g in Giorni } :
  attivazioneGiornaliera[t, f, g] <= attivazioneSettimanale[f];
subject to attivazioniGiornaliere { t in Tipo, f in Fornitori, g in Giorni
  } : (ambulanze[t, f, g] + ambulanzeSurplus[t, f, g]) <= BigM *
  attivazioneGiornaliera[ t, f, g];

# Vincolo che Limita l'Attivazione dei Fornitori nell'Arco della Settimana
  (Considerando il Tipo di Ambulanza) (Vincolo 1)
subject to massimaAttivazione { t in Tipo, f in Fornitori } : sum{ g in
  Giorni } attivazioneGiornaliera[t, f, g] <= maxGiorni[t, f];

```

---

## 4 Scenari

**Premessa:** dato che negli scenari proposti gli insiemi non variano, questi saranno di seguito presentati una sola volta. Gli insiemi sono così definiti:

Giorni	lunedì	martedì	mercoledì	giovedì	venerdì	sabato	domenica
--------	--------	---------	-----------	---------	---------	--------	----------

Fornitori	interna	bianca	rossa	verde
-----------	---------	--------	-------	-------

Tipo	trasporto	emergenza
------	-----------	-----------

### 4.1 Primo Scenario

#### 4.1.1 Descrizione

In questo primo scenario, i dati sono stati inseriti manualmente, con un numero di ambulanze necessarie tale da permettere di trovare una soluzione ottima e che sfrutti tutti i vincoli imposti.

#### 4.1.2 Dati

Bisogno			Surplus		
Giorno	Trasporto	Emergenza	Giorno	Trasporto	Emergenza
lunedì	3	20	lunedì	1	3
martedì	5	22	martedì	3	3
mercoledì	2	10	mercoledì	1	3
giovedì	3	20	giovedì	1	2
venerdì	4	30	venerdì	2	5
sabato	2	30	sabato	1	6
domenica	0	15	domenica	1	4

maxAmbulanze			maxGiorni		
OdV	Trasporto	Emergenza	OdV	Trasporto	Emergenza
interna	2	3	interna	7	7
bianca	5	0	bianca	3	0
rossa	0	15	rossa	0	5
verde	10	30	verde	4	6

costoGiornaliero			addSurplus	
OdV	Trasporto	Emergenza	OdV	Fattore Moltiplicativo
interna	5	8	interna	0.00
bianca	10	0	bianca	5.00
rossa	0	10	rossa	8.00
verde	7	15	verde	10.00

costoAttivazioneSettimanale	
OdV	Costo
interna	25
bianca	150
rossa	130
verde	170

#### 4.1.3 Output

L'esecuzione del file `.run` relativo a questo scenario mostra che il costo minimo sostenibile è di €3595. Come si può vedere dall'output, ogni giorno sono stati attivati almeno 2 fornitori per tipo di ambulanza. Oltre a questo si può notare che, per quanto riguarda le ambulanze surplus, il programma ha scelto giustamente di usare quelle interne in quanto il fattore moltiplicativo per tale utilizzo è nullo.



## 4.2 Secondo Scenario

### 4.2.1 Descrizione

Il secondo set di dati prevede un bisogno e un surplus impossibili da soddisfare, per questo già il *presolve* ritornerà un errore che indica l'impossibilità di trovare una soluzione con i vincoli di riferimento. Senza riportare nuovamente le tabelle, il set di dati utilizza tutti i dati di default tranne che per i parametri *bisogno* e *surplus* che hanno un valore di 200.

### 4.2.2 Output

Ovviamente, il costo minimo trovato eseguendo il file **.run** sarà pari a €0. La non risolubilità del problema è data dal fatto che l'insieme delle ambulanze dedicate alle emergenze (20) e dedicate al trasporto (20) non sono abbastanza per coprire il fabbisogno dell'intera settimana (ossia di 200 ambulanze per tipo al giorno).

## 4.3 Terzo Scenario

### 4.3.1 Descrizione

Nel terzo scenario sono stati definiti soltanto gli insiemi, così da poter sfruttare i valori di default. Per questo motivo non verrà riportato il valore dei vari parametri in quanto specificato nel modello.

### 4.3.2 Output

Eseguendo il file **.run** dedicato, utilizzando quindi i dati di default, il costo minimo risulta essere di €2720.

## 5 Extra

### 5.1 Output su File

All'interno della cartella **AMPL** si può trovare un file denominato *Ospedale.GenerateOutputFile.run* che esegue in serie tutti e tre gli scenari sopracitati e, tramite un secondo file *.run* di supporto denominato *GenerateOutput.run*, ne salva l'output in un file dedicato chiamato *Output.txt*. Per fare ciò, sono stati utilizzati gli operatori `>` e `>>`. Il primo serve a creare un nuovo file (o a ripulirlo se già esistente) e scriverci l'istruzione della stessa linea; il secondo serve invece per aprire il file e fare un *append* dell'istruzione di riferimento.

### 5.2 Stampa Personalizzata per Problema Non Risolvibile

All'interno dei file *Ospedale.xxx.run* dove *xxx* indica il numero del caso preso in esame o il file dedicato alla stampa, c'è un *if-then-else statement* che, utilizzando il valore della variabile di stato *solve\_result*, determina se un problema è risolvibile o meno e modifica l'output di conseguenza. Nel caso il problema non fosse risolvibile, tramite la variabile *solve\_result\_num*, indica il codice che identifica la non risolvibilità del problema.